
Abstract

The worker/wrapper transformation is a technique for changing the type of a computation,
usually with the aim of improving its performance. It has been used by compiler writers for
many years, but the technique is little-known in the wider functional programming com-
munity, and has never been described precisely. In this article we explain, formalise, and
explore the generality of the worker/wrapper transformation. We also provide a systematic
recipe for its use as an equational reasoning technique for improving the performance of
programs, and illustrate the power of this recipe using a range of examples.

1 Introduction

The worker/wrapper transformation is a technique for transforming a computation
of one type into a worker of a different type, together with a wrapper that acts as
an impedance matcher between the original and new computations.

The technique can be used to improve the performance of functional programs by
improving the choice of data structures used. For example, in the Glasgow Haskell
Compiler, a specific instance of the worker/wrapper transformation is used to re-
place boxed data structures by unboxed structures, based upon strictness informa-
tion (Peyton Jones & Launchbury, 1991). Despite its practical importance, however,
the technique is little-known in the wider functional programming community, and
to the best of our knowledge has never been described precisely.

In this article we explain, formalise, and explore the generality of the worker/
wrapper transformation. We also provide a systematic recipe for its use, and il-
lustrate the power of this recipe using a range of programming examples. More
precisely, the article makes the following contributions:

• We present the worker/wrapper transformation in a general semantic frame-
work (section 3), rather than via specific syntactic instances, in order to make
the technique as widely applicable as possible.

• We give a correctness proof for the technique (section 3), based upon the use
of the rolling rule from fixed point calculus, which provides a range of explicit
conditions under which the technique can be applied.

• We provide a systematic recipe for its use as an equational reasoning technique
for improving the performance of programs (section 3), by means of a step-
by-step process for applying the transformation.

• We apply this recipe to four different approaches to program optimisation,
based upon the use of accumulation (section 4), unboxing (section 5), mem-
oisation (section 6), and continuations (section 7).

A summary of related work, and directions for further work, are provided in
the concluding sections. The article is aimed at readers who are familiar with the
basics of reasoning about functional programs, say to the level of (Bird, 1998),
but no previous experience with the worker/wrapper transformation is assumed.
The techniques are presented using Haskell (Peyton Jones, 2003), but can easily be
adapted to other functional languages.

2 The basic idea

In this section we review the basic idea of the worker/wrapper transformation,
in a similar manner to which it was originally described by Peyton Jones and
Launchbury (1991), and implemented in the Glasgow Haskell Compiler.

Suppose that we are given a function f , defined in the form

f = body

where body is the right-hand side of the definition, and may include recursive calls
to f . At this syntactic level, the first step in applying the worker/wrapper transfor-
mation is to define appropriate functions wrap and unwrap that allow the function
f to be redefined by the equation f = wrap (unwrap body), which is then split into
two equations by naming the intermediate result, as follows:

f = wrap work

work = unwrap body

In this manner, f has been factorised into the application of a “wrapper” function
wrap to a “worker” function work , itself defined by applying unwrap to the body
of the original definition for f . Note that if f was originally recursive, then f and
work are now mutually recursive. The next step in the process is to eliminate such
mutual recursion by inlining the new definition for f in the definition for work ,
thereby making work into a recursive definition that is independent of f :

f = wrap work

work = unwrap (body [wrap work / f ])

As usual, e [e′/x] denotes the result of substituting e ′ for all free occurrences of
the variable x in the expression e. The final step is then to simplify the resulting
definitions for f and work using standard program transformation techniques.

For example, Peyton Jones and Launchbury (1991) show how this process can be
used to transform a recursive definition of the factorial function into a more efficient

version that is defined in terms of a worker that only uses unboxed integers, together
with a wrapper that takes care of the initial unboxing and final boxing.

The above, syntactic, description of the worker/wrapper transformation is ap-
pealingly simple, but raises a number of important questions. Is the technique
actually correct? How can this be proved? Under what conditions does the proof
hold? How should it be used in practice? What kind of applications is it suitable
for? Addressing these questions is the purpose of the remainder of this article.

3 The worker/wrapper transformation

In order to give a precise treatment of the worker/wrapper transformation, and
hence prove its correctness, we move from the informal syntactic approach based
upon inlining to a formal semantic approach based upon the use of fixed points.
We begin by defining a fixed point operator in Haskell:

fix :: (a → a) → a
fix f = f (fix f )

The property of this operator that we will use to formalise the worker/wrapper
transformation is the rolling rule (Backhouse, 2002), which allows us to pull the
first argument of a composition outside a fixed point, resulting in the composition
swapping the order of its arguments, or “rolling over”:

fix (g ◦ f ) = g (fix (f ◦ g))

Informally, this rule is valid because both sides expand to the infinite application
g (f (g (f (g (f · · ·))))). A formal proof, together with a brief review of the fixed
point approach to recursion, is provided in the appendix.

Now consider the problem of changing the type of a recursive computation. More
precisely, suppose that we are given a computation comp :: A, defined as the least
fixed point comp = fix body of some function body :: A → A, and that we wish to
change the underlying type from A to some other type B .

The worker/wrapper approach to this problem of changing types is based upon
defining conversion functions unwrap::A → B and wrap ::B → A such that the equa-
tion wrap ◦ unwrap = idA holds, where idA is the identity function for the type A.
(To aid the reader, identity functions are subscripted with their types throughout
the article.) This equation states that converting a value of the original type into
the new type and then back again does not change the value, and formalises the
idea that the type A can be faithfully represented by the type B .

Given such a setting, we can now calculate as follows:

comp
= { applying comp }

fix body
= { id is the identity for ◦ }

fix (idA ◦ body)
= { † assuming wrap ◦ unwrap = idA }

fix (wrap ◦ unwrap ◦ body)
= { rolling rule }

wrap (fix (unwrap ◦ body ◦ wrap))
= { define work = fix (unwrap ◦ body ◦ wrap) }

wrap work

That is, we have shown how an arbitrary recursive computation of type A can
be factorised as a wrapper of type B → A applied to a new recursive worker
computation of type B , based upon the assumption that the identity function for
A can be split into the composition of two conversion functions. The following
diagram summarises the primary typing relationships:

comp :: A

unwrap

��
work :: B

wrap

��

As we shall see, however, in general there is no requirement that this diagram
commutes. For example, for some applications the assumption wrap ◦unwrap = idA

may not be true in general, but only in the context (labelled by †) in which it is
used in the above calculation. In particular, we may sometimes require the weaker
property wrap ◦unwrap ◦body = idA◦body , which states that the assumption is true
for values produced by body , or even fix (wrap ◦ unwrap ◦ body) = fix (idA ◦ body),
which also takes the recursive context into account. That is, we have the following
hierarchy of worker/wrapper assumptions that support the above calculation:

wrap ◦ unwrap = idA

⇓
wrap ◦ unwrap ◦ body = body

⇓
fix (wrap ◦ unwrap ◦ body) = fix body

For any given application, we will use the strongest such assumption that is valid, i.e.
the simplest to verify; often the basic equation wrap ◦ unwrap = idA will suffice. In
conclusion, the original syntactic description of the worker/wrapper transformation
can now be formalised on semantic grounds as shown in Figure 1.

Once this transformation has been applied, one then attempts to simplify the
worker using normal fold/unfold transformation (Burstall & Darlington, 1977). It
is usually convenient to begin by rewriting the worker using explicit recursion

work = unwrap (body (wrap work ))

which transformation can be verified as follows:

work

If comp :: A is a recursive computation defined by comp = fix body for
some body ::A → A, and wrap ::B → A and unwrap ::A → B are conversion
functions satisfying any of the worker/wrapper assumptions, then

comp = wrap work

where work :: B is defined by

work = fix (unwrap ◦ body ◦ wrap)

Fig. 1. The worker/wrapper transformation

= { applying work }
fix (unwrap ◦ body ◦ wrap)

= { applying fix }
(unwrap ◦ body ◦ wrap) (fix (unwrap ◦ body ◦ wrap))

= { unapplying work }
(unwrap ◦ body ◦ wrap) work

= { applying ◦ }
unwrap (body (wrap work ))

Further simplification of the worker is then typically driven by the desire to fuse
together instances of unwrap and wrap, to eliminate the overhead of repeatedly
converting between the two types. In general, it is not the case that unwrap ◦wrap
can be fused to give idB, but the following worker/wrapper fusion property, in
which the argument value of type B is the worker itself, is often applicable:

If wrap ◦ unwrap = idA, then unwrap (wrap work ) = work

For example, this property will be utilised in two of our programming applications
(sections 4 and 7). The fusion property itself can be verified as follows:

unwrap (wrap work )
= { applying work }

unwrap (wrap (unwrap (body (wrap work ))))
= { assuming wrap ◦ unwrap = idA }

unwrap (body (wrap work ))
= { unapplying work }

work

In summary, we have the following general recipe for applying the worker/wrapper
transformation to change the type of a recursive computation:

• Express the computation as a least fixed point;

• Choose the desired new type for the computation;

• Define conversions between the original and new types;

• Check that they satisfy one of the worker/wrapper assumptions;

• Apply the worker/wrapper transformation;

• Simplify the resulting definitions.

We conclude this section by noting that the worker/wrapper transformation can
also be formalised using fixed point fusion (Meijer et al., 1991) which requires the
additional property that wrap is strict (wrap ⊥ = ⊥). Fortunately this property
follows automatically from the basic worker/wrapper assumption wrap ◦ unwrap =
id , by virtue of the fact that any (monotonic) function that is right invertible must
be strict. We prefer the formulation using the rolling rule because it corresponds
directly to our understanding of why the transformation is valid.

4 Example: the reverse function

As a first example of our worker/wrapper recipe, we will use it to transform a simple
definition for the function that reverses a list into a more efficient version that uses
accumulation. This transformation is normally achieved using more elementary
techniques (Hutton, 2007, section 13.6), but we now show that it also fits naturally
into our general worker/wrapper paradigm.

4.1 Hughes lists

We begin by reviewing the idea of representing lists using functions, which is the
essential algorithmic idea underlying this example.

As observed by Hughes (1986), it is possible to represent the normal type of lists
in an alternative manner using functions, which represent a list xs by the function
(xs ++) that prepends xs to its argument. We can implement this idea by defining
a type H a of Hughes lists, together with a representation function rep:

type H a = [a ] → [a ]

rep :: [a ] → H a
rep xs = (xs ++)

An important property of this representation is that the function rep forms a ho-
momorphism from lists to functions, in the sense that:

rep (xs ++ ys) = rep xs ◦ rep ys

rep [ ] = id[a]

That is, rep is a homomorphism from the monoid of (finite) lists, for which the
associative operator is ++ and the unit is [ ], to the monoid of functions on lists,
for which the operator is composition ◦ and the unit is the identity function id[a].
These two equations can be verified by straightforward calculation:

rep (xs ++ ys) zs
= { applying rep }

(xs ++ ys) ++ zs
= { associativity of ++ }

xs ++ (ys ++ zs)
= { unapplying rep }

rep xs (ys ++ zs)
= { unapplying rep }

rep xs (rep ys zs)
= { unapplying ◦ }

(rep xs ◦ rep ys) zs

and

rep [ ] ys
= { applying rep }

[ ] ++ ys
= { applying ++ }

ys

In addition to the function rep that converts normal lists into Hughes lists, it is
also natural to consider conversion in the opposite direction, which is achieved by
simply applying the given function to the empty list:

abs :: H a → [a ]
abs f = f [ ]

This definition ensures that converting a list into Hughes form and back again
returns the original list, or more formally, it ensures abs is the left inverse to rep,

abs ◦ rep = id[a]

which can be verified as follows:

(abs ◦ rep) xs
= { applying ◦ }

abs (rep xs)
= { applying abs }

rep xs [ ]
= { applying rep }

xs ++ [ ]
= { [ ] is the identity for ++ }

xs

In summary, using the terminology of data representation (Hoare, 1972), we have
shown how the ‘abstract’ type [a ] can be represented using the ‘concrete’ type H a.
As is usual in this setting, note that the representation function rep is injective but
not surjective, while the abstraction function abs is surjective but not injective.

4.2 Reverse

Now consider the following definition for the function that reverses a list:

rev :: [a ] → [a ]
rev [ ] = [ ]
rev (x : xs) = rev xs ++ [x ]

Because of the use of ++, this definition takes quadratic time in the length of
its argument. We now show how this definition can be transformed into a more
efficient worker that uses an extra argument to accumulate the result, together
with a wrapper that takes care of the initial setup. The first step is to redefine rev
as a least fixed point, by abstracting over the recursive call in the body:

rev :: [a ] → [a ]
rev = fix body

body :: ([a ] → [a ]) → ([a ] → [a ])
body f [ ] = [ ]
body f (x : xs) = f xs ++ [x ]

Following our worker/wrapper recipe, and utilising the idea of Hughes lists, our aim
now is to transform the computation rev of type [a ] → [a ] into a worker of type
[a ] → H a (which abbreviates [a ] → [a ] → [a ] and hence introduces the necessary
extra argument list), as illustrated by the following diagram:

rev :: [a ] → [a ]

unwrap

��
work :: [a ] → H a

wrap

��

Implementing the conversion functions unwrap and wrap is simply a matter of
composing the conversion functions rep and abs from the previous section:

unwrap :: ([a ] → [a ]) → ([a ] → H a)
unwrap f = rep ◦ f

wrap :: ([a ] → H a) → ([a ] → [a ])
wrap g = abs ◦ g

Verifying the worker/wrapper assumption, in this case wrap ◦unwrap = id[a]→[a],
follows from the fact that abs ◦ rep = id[a] by straightforward calculation:

(wrap ◦ unwrap) f
= { applying ◦ }

wrap (unwrap f )
= { applying wrap }

abs ◦ unwrap f
= { applying unwrap }

abs ◦ rep ◦ f
= { abs ◦ rep = id[a] }

f

Now that we have satisfied the preconditions for the worker/wrapper transforma-
tion, applying this transformation and rewriting the worker using explicit recursion
rather than the fix operator gives the following result:

rev :: [a ] → [a ]
rev = wrap work

work :: [a ] → H a
work = unwrap (body (wrap work ))

Making the list argument explicit in the definition for rev , and expanding out
wrap, we obtain the wrapper for the efficient version of rev , which simply supplies
the empty list as the initial accumulator:

rev :: [a ] → [a ]
rev xs = work xs [ ]

Now let us focus our attention on simplifying the definition for the worker func-
tion. First of all, we redefine work using pattern matching

work [ ] = rep [ ]
work (x : xs) = rep (wrap work xs ++ [x ])

which transformation can be verified as follows:

work xs
= { applying work }

unwrap (body (wrap work )) xs
= { applying unwrap }

rep (body (wrap work ) xs)
= { applying body }

rep (case xs of
[ ] → [ ]
(x : xs) → wrap work xs ++ [x ] )

= { distribution over case }
case xs of

[ ] → rep [ ]
(x : xs) → rep (wrap work xs ++ [x ])

The distribution step assumes that rev is only applied to finite lists, to ensure that
the argument list xs in this step is not undefined (⊥).

Now we exploit the fact that rep :: [a ] → H a is a homomorphism from lists to
functions, to rewrite the new definition as follows:

work [ ] = id[a]

work (x : xs) = rep (wrap work xs) ◦ rep [x ]

Then we simplify the first argument of the composition in this definition by exposing
the implicit use of unwrap and then fusing this with wrap:

rep (wrap work xs)
= { unapplying unwrap }

unwrap (wrap work ) xs
= { worker/wrapper fusion }

work xs

That is, we now have the following definition:

work [ ] = id[a]

work (x : xs) = work xs ◦ rep [x ]

Finally, if we make the extra list argument in this definition explicit, and then
expand out ◦ and rep, we obtain the worker for the efficient version of reverse,
which uses an extra list argument to accumulate the final result:

work :: [a ] → [a ] → [a ]
work [ ] ys = ys
work (x : xs) ys = work xs (x : ys)

We conclude this section with three observations about the above derivation.
First of all, note that everything follows in a straightforward manner from the idea
of using Hughes’ representation of lists, in the sense that once we have made this
decision, the rest of the derivation proceeds using standard fold/unfold transforma-
tion. Secondly, the derivation does not require the use of induction, other than the
implicit use to verify the auxiliary result that finite lists form a monoid, which is
used in proof that rep is a homomorphism. And finally, it is possible to generalise
from this example and formulate a special case of the worker/wrapper transforma-
tion for changing the result type of a function; because this only saves a few steps
in the above calculations, we prefer to use our general formulation.

5 Example: the factorial function

For our second example, we formalise an example from the original article on the
worker/wrapper transformation (Peyton Jones & Launchbury, 1991), by showing
how to transform a simple definition for the factorial function on integers into a
more efficient version that only uses unboxed integers.

5.1 Boxed and unboxed integers

Consider the question of how integers should be represented in the implementa-
tion of a programming language. For strict languages, such as ML, integers can be
represented directly as binary numbers, for example as a 32-bit word. Integers repre-
sented in this manner are called unboxed integers, because no additional packaging
is required around the raw numeric values themselves.

For non-strict languages, such as Haskell, a more refined representation is re-
quired, because evaluation is only performed on demand. In particular, we must
distinguish between a computation that will return an integer value, and an integer
value itself. In practice, this is usually achieved using some form of tagging in the
implementation. Such tagged integer representations are called boxed integers.

Tagging supports the use of non-strict evaluation, but carries a considerable over-
head. For example, evaluating x +y conceptually requires evaluating the expressions
x and y, unboxing the resulting values (removing the tags), performing the actual
addition, and then boxing the result (reinstating the tag). It would clearly be more
efficient to work with unboxed integers as much as possible.

The key to achieving this behaviour in Haskell is an idea put forward by Peyton
Jones and Launchbury (1991), namely to reflect the notion of boxed and unboxed
integers directly in the language itself, via the following type definition:

data Int = I# Int#

That is, a boxed value of type Int can be constructed by applying the constructor
tag I# to a value of type Int#, which is the built-in type of unboxed 32-bit integers,
and is unlifted in the sense that it has no ⊥ element. Note that, by convention
in Haskell, unboxed types, values, variables and functions have a # suffix on their
name, as in Int#, 0#, n#, and +#. However, this symbol has no semantic meaning,
and is used purely for identification purposes.

Given the above type definition, converting an unboxed integer into a boxed
integer can be achieved simply by applying the constructor I#, and in the opposite
direction by removing this constructor through pattern matching. Thus there is no
need to define explicit conversion functions.

5.2 Factorial

Consider the following definition for the factorial function on non-negative integers:

fac :: Int → Int
fac 0 = 1
fac n = n ∗ fac (n − 1)

In a non-strict language such as Haskell, a straightforward implementation of this
definition will be rather inefficient, due to repeated unboxing and boxing of integer
values. However, because fac is strict (fac ⊥ = ⊥), it is safe to replace the defini-
tion of fac by a more efficient worker that only uses unboxed integers and strict
evaluation, together with a wrapper that takes care of the initial unboxing and final
reboxing. We show how this transformation can be achieved in practice. Following

our recipe, the first step is to redefine fac as a least fixed point:

fac :: Int → Int
fac = fix body

body :: (Int → Int) → (Int → Int)
body f 0 = 1
body f n = n ∗ f (n − 1)

Our aim now is to transform the computation fac of type Int → Int into a worker
of type Int# → Int#, as illustrated in the following diagram:

fac :: Int → Int

unwrap

��
work :: Int# → Int#

wrap

��

Implementing the conversion functions wrap and unwrap is simply a matter of
taking care of the necessary boxing and unboxing:

unwrap :: (Int → Int) → (Int# → Int#)
unwrap f x# = case f (I# x#) of

I# y# → y#

wrap :: (Int# → Int#) → (Int → Int)
wrap g# x = case x of

I# x# → case g# x# of
y# → I# y#

Using the above definitions and properties of case expressions, it is straightfor-
ward to show that wrap ◦ unwrap = idInt→Int , but only if we restrict our attention
to argument functions of type Int → Int that are strict:

(wrap ◦ unwrap) f x
= { applying ◦ }

wrap (unwrap f ) x
= { applying wrap }

case x of
I# x# → case unwrap f x# of

y# → I# y#

= { applying unwrap }
case x of

I# x# → case (case f (I# x#) of
I# z# → z# ) of

y# → I# y#

We now proceed by case analysis on x :: Int .

Case: x = ⊥

case ⊥ of · · ·
= { case is strict }

⊥
= { assuming f is strict }

f ⊥

Case: x = I# x# for some x# :: Int#

case I# x# of
I# x# → case (case f (I# x#) of

I# z# → z# ) of
y# → I# y#

= { case of known constructor }
case (case f (I# x#) of

I# z# → z# ) of
y# → I# y#

= { case of case }
case f (I# x#) of

I# z# → case z# of
y# → I# y#

= { eliminating the inner case }
case f (I# x#) of

I# z# → I# z#
= { case elimination }

f (I# x#)

Hence for this example, the simple assumption wrap ◦ unwrap = idInt→Int is not
always valid. However, because body f is defined by pattern matching and is there-
fore strict, we do have the weaker assumption wrap ◦ unwrap ◦ body = body , which
is precisely where strictness is exploited in this example. More complex functions
than factorial, for which determining strictness depends upon taking account of
recursion, will require our weakest worker/wrapper assumption defined using fix .

Now that we have satisfied the worker/wrapper preconditions, applying this
transformation gives the following result:

fac :: Int → Int
fac = wrap work

work :: Int# → Int#
work = unwrap (body (wrap work ))

Making the integer argument explicit in the definition for fac, and expanding out
wrap, we obtain the wrapper for the unboxed version of fac, which unboxes the



14 A. Gill and G. Hutton

integer argument, applies the worker, and then boxes the result:

fac :: Int → Int
fac n = case n of

I# x# → case work x# of
y# → I# y#

Now let us simplify the definition for the worker function, which in this case
amounts to expanding out definitions and then simplifying the result. As with the
reverse example, the first step is to redefine work using pattern matching:

work x#

= { applying work }
unwrap (body (wrap work )) x#

= { applying unwrap }
case body (wrap work ) (I# x#) of

I# y# → y#

= { applying body }
case (case (I# x#) of

0 → 1
n → n ∗ wrap work (n − 1)) of

I# y# → y#

= { applying wrap }
case (case (I# x#) of

0 → 1
n → n ∗ (case (n − 1) of

I# a# → case work a# of
b# → I# b# )) of

I# y# → y#

= { distribution over case, n∗ is strict }
case (case (I# x#) of

0 → 1
n → case (n − 1) of

I# a# → case work a# of
b# → n ∗ (I# b#)) of

I# y# → y#

Now we expand out the boxed values 0 and 1, together with the boxed variable n,
in terms of the constructor I#, to rewrite the new definition as follows:

work x# = case (case (I# x#) of
I# 0# → I# 1#

I# n# → case ((I# n#) − (I# 1#)) of
I# a# → case work a# of

b# → (I# n#) ∗ (I# b#)) of
I# y# → y#

Then we expand out the subtraction − and multiplication ∗ operators on boxed

integers, using the fact that any such operator ⊕ is defined by

n ⊕ m = case n of
I# x# → case m of

I# y# → case (x# ⊕# y#) of
z# → I# z#

to give the following result:

work x# = case (case (I# x#) of
I# 0# → I# 1#

I# n# → case (case n# −# 1# of
c# → I# c#) of

I# a# → case work a# of
b# → case n# ∗# b# of

d# → I# d#) of
I# y# → y#

Finally, if we simplify this definition using properties of case, we obtain a worker
with the same structure as the original definition for the factorial function, but
which now operates entirely using unboxed integers, and hence avoids the need for
repeated unboxing and boxing of integer values:

work :: Int# → Int#
work x# = case x# of

0# → 1#

n# → n# ∗# work (n# −# 1#)

Once again, note that once we have made the initial decision regarding the change
in type of the function, applying our worker/wrapper recipe is largely a matter of
routine equational reasoning, and does not require the use of induction. Note also
that this derivation involved changing both the argument and result type of a
function, whereas our previous example only changed the result type.

6 Example: the Fibonacci function

For our next example, we show how the worker/wrapper transformation can be used
to transform a simple definition for the Fibonacci function on natural numbers into
a more efficient version that operates by means of a memo table. In contrast to our
previous two examples, in which a function is represented by another function, this
derivation represents a function by a data structure.

6.1 Memoisation

The term memoisation was coined by Michie (1968), and refers to the optimisation
technique of tabulating all the arguments that a function is called with, together
with the corresponding results returned, and reusing these results if the function is

called again with any of its previous arguments. If this happens often, memoisation
can lead to a dramatic improvement in performance.

Our approach to memoising the Fibonacci function is to observe that any function
on natural numbers can be represented in an alternative manner as an infinite
list (or stream) of results. In particular, we represent a function f as the stream
[f 0, f 1, f 2, · · ·] that tabulates its behaviour for all possible argument values.

To implement this representation, let us write Nat for the subtype of Int compris-
ing the natural numbers 0, 1, 2, 3, · · ·, and Stream a for the subtype of [a ] comprising
the infinite lists of elements of type a. Using these two subtypes allows us to be
more precise about the types of functions that we will define, while still being able
to use familiar notation and library functions for integers and lists.

For example, a function unwrap that converts functions on natural numbers into
streams in the manner described above can now be defined:

unwrap :: (Nat → a) → Stream a
unwrap f = map f [0 . .]

For the purposes of proofs, however, an equivalent definition that uses explicit co-
recursion (Gibbons & Hutton, 2005) is usually preferable:

unwrap f = f 0 : unwrap (f ◦ (+1))

Dually, we can also perform conversion in the opposite direction, from a stream
to a function, by simply indexing into the stream:

wrap :: Stream a → (Nat → a)
wrap xs = (xs !!)

The auxiliary operator !! selects the nth element of a stream:

(!!) :: Stream a → Nat → a
xs !! 0 = head xs
xs !! (n + 1) = (tail xs) !! n

Using these definitions, we can show that wrap ◦ unwrap = idNat→a by making
the function and numeric arguments explicit, then expanding out definitions to give
(unwrap f ) !! n = f n, and finally verifying this equation by induction on n.

Base case:

(unwrap f ) !! 0
= { applying unwrap }

(f 0 : unwrap (f ◦ (+1))) !! 0
= { applying !! }

f 0

Inductive case:

(unwrap f ) !! (n + 1)
= { applying unwrap }

(f 0 : unwrap (f ◦ (+1))) !! (n + 1)

= { applying !! }
(unwrap (f ◦ (+1))) !! n

= { induction hypothesis }
(f ◦ (+1)) n

= { applying ◦ }
f (n + 1)

The dual property, unwrap ◦wrap = idStream a, also holds. In particular, making
the stream argument explicit and expanding out definitions gives unwrap (xs !!) =
xs, which can be verified by coinduction on streams (Turner, 1995):

unwrap (xs !!)
= { applying unwrap }

(xs !! 0) : unwrap ((xs !!) ◦ (+1))
= { applying !! }

head xs : unwrap ((xs !!) ◦ (+1))
= { applying ◦ }

head xs : unwrap (λn → xs !! (n + 1))
= { applying !! }

head xs : unwrap (λn → (tail xs) !! n)
= { sections }

head xs : unwrap ((tail xs) !!)
= { coinduction hypothesis }

head xs : tail xs
= { streams }

xs

In conclusion, the functions unwrap and wrap establish an isomorphism between
the types Nat → a and Stream a. This isomorphism itself is well-known, but
perhaps surprisingly, generalises in allowing any function type with an inductively-
defined argument type to be represented in a coinductive manner (Altenkirch, 2001),
thereby providing a basis for memoising a large class of functions.

6.2 Fibonacci

Consider the following standard definition for the Fibonacci function on natural
numbers, as found in any textbook on numerical series:

fib :: Nat → Nat
fib 0 = 0
fib 1 = 1
fib (n + 2) = fib n + fib (n + 1)

That is, the first two Fibonacci numbers are 0 and 1, and each successive number
is the sum of the previous two. While this definition directly expresses the desired
behaviour of the function, it also recomputes the same results many times over, in
fact requiring exponential time in the size of its argument. We now show how it

can be transformed into a more efficient worker that constructs a memo table com-
prising the stream [0, 1, 1, 2, 3, 5, · · ·] of all Fibonacci numbers (and hence ensures
that each result is computed at most once), together with a wrapper that selects
the required element from this table. The first step is to redefine fib using fix :

fib :: Nat → Nat
fib = fix body

body :: (Nat → Nat) → (Nat → Nat)
body f 0 = 0
body f 1 = 1
body f (n + 2) = f n + f (n + 1)

Our aim now is to transform the computation fib of type Nat → Nat into a
worker of type Stream Nat , as illustrated in the following diagram:

fib :: Nat → Nat

unwrap

��
work :: Stream Nat

wrap

��

Because we have already defined appropriate conversion functions satisfying the
assumption wrap ◦ unwrap = idNat→Nat in the previous section, we can now apply
the worker/wrapper transformation to obtain the following result:

fib :: Nat → Nat
fib = wrap work

work :: Stream Nat
work = unwrap (body (wrap work ))

Making the numeric argument explicit in the definition for fib, and expanding out
wrap, we obtain the wrapper for the memoised version of fib, which simply selects
the required element from the memo table:

fib :: Nat → Nat
fib n = work !! n

Now we simplify the worker itself, which in this case just amounts to expanding
out definitions, and does not require any form of fusion:

work
= { applying work }

unwrap (body (wrap work ))
= { applying unwrap }

map (body (wrap work )) [0 . .]
= { applying body }

map (λn → case n of

0 → 0
1 → 1
(n + 2) → wrap work n + wrap work (n + 1)) [0 . .]

= { applying wrap }
map (λn → case n of

0 → 0
1 → 1
(n + 2) → work !! n + work !! (n + 1)) [0 . .]

That is, we have derived the following definition for the worker, which builds a
memo table comprising all Fibonacci numbers, using the table itself to ensure that
each number is computed at most once, and lazy evaluation to ensure that the table
(which is conceptually infinite) is built on a demand-driven basis:

work :: Stream Nat
work = map f [0 . .]

where
f 0 = 0
f 1 = 1
f (n + 2) = work !! n + work !! (n + 1)

In conclusion, we have improved the time performance of the fib function from
exponential to quadratic. Further improvements can readily be made in a variety of
ways, such as representing the memo table in a more efficient manner (e.g. using an
array with constant-time indexing rather than a stream with linear-time indexing),
exploiting the structure of the computation to design a specialised version of the
memo table (e.g. using a pair of numbers rather than a stream), or using more
advanced programming techniques (e.g. cyclic programming). For example, using
the technique of unique fixed points (Hinze, 2008), it can be shown that the above
definition for the worker is equivalent to the well-known cyclic definition for the
stream of Fibonacci numbers:

fibs :: Stream Nat
fibs = 0 : 1 : zipWith (+) fibs (tail fibs)

7 Example: an evaluation function

For our final example, we show how to use the worker/wrapper transformation in
implementing the transformation of a function into continuation-passing style, an
important precursor to many program analyses and optimisations.

7.1 Continuations

As observed by Reynolds (1972), it is possible to represent any type in an alternative
manner using continuations . The notion of a continuation can be defined in many
ways, but for our purposes it is simply a function that is applied to the result
of another computation. Using this idea, we represent a value x as the function

λc → c x that takes another function c (a continuation) as its argument, and
applies this function to x in order to produce the final result.

We can implement this representation using the type (a → r) → r of continuation
computations of type a that return results of type r (Wadler, 1992b). For our
purposes, however, we only require the special case when the two types are the
same, which we abbreviate by Cont a, and define a function rep that converts
normal values into their representation as continuation values:

type Cont a = (a → a) → a

rep :: a → Cont a
rep x = λc → c x

Dually, we can also perform conversion in the opposite direction, by supplying the
identity function as the continuation:

abs :: Cont a → a
abs f = f ida

Using these definitions, it is easy to show that abs ◦ rep = ida:

(abs ◦ rep) x
= { applying ◦ }

abs (rep x )
= { applying rep }

abs (λc → c x )
= { applying abs }

(λc → c x ) ida

= { beta reduction }
ida x

= { applying id }
x

7.2 Evaluation

Consider a simple expression language comprising integers values, an addition op-
erator, a single exceptional value called throw, and a catch operator:

data Expr = Val Int | Add Expr Expr | Throw | Catch Expr Expr

This language provides an appropriate minimal setting in which to investigate var-
ious aspects of the semantics of exceptions (Hutton & Wright, 2004; 2006; 2007).
Informally, Throw abandons the current computation and throws an exception,
while Catch x y behaves as the expression x unless it throws an exception, in
which case the catch behaves as the handler expression y.

To formalise the meaning of this language, we first recall the Maybe type:

data Maybe a = Nothing | Just a

That is, a value of type Maybe a is either Nothing, which we think of as an ex-
ceptional value, or has the form Just x for some x of type a, which we think of as

a normal value (Spivey, 1990). For the purposes of our example, however, we only
require the type Maybe Int , which we abbreviate as Mint :

type Mint = Maybe Int

Using this type, it is straightforward to define a function that evaluates expressions,
and takes care of the necessary propagation and handling of exceptions:

eval :: Expr → Mint
eval (Val n) = Just n
eval (Add x y) = case eval x of

Nothing → Nothing
Just n → case eval y of

Nothing → Nothing
Just m → Just (n + m)

eval (Throw ) = Nothing
eval (Catch x y) = case eval x of

Nothing → eval y
Just n → Just n

Because of the repeated tagging and untagging of values of type Mint , this definition
may be rather inefficient. We now show how it can be transformed into a worker
that uses two continuations to make normal and exceptional control flow explicit
(and hence eliminate the use of tags), together with a wrapper that takes care of
the initial setup. Once again, the first step is to redefine eval using fix :

eval :: Expr → Mint
eval = fix body

body :: (Expr → Mint) → (Expr → Mint)
body f (Val n) = Just n
body f (Add x y) = case f x of

Nothing → Nothing
Just n → case f y of

Nothing → Nothing
Just m → Just (n + m)

body f (Throw ) = Nothing
body f (Catch x y) = case f x of

Nothing → f y
Just n → Just n

One might expect now to aim to transform the computation eval of type Expr →
Mint into a worker of type Expr → Cont Mint . In practice, however, the desire
to have separate continuations for normal and exceptional control flow, sometimes
called double-barrelled continuations (Thielecke, 2002), means that we start by split-
ting the underlying type Mint → Mint of continuations into two parts, and aim for
a worker of the type illustrated in the following diagram:

eval :: Expr → Mint

unwrap

��
work :: Expr → (Int → Mint) → Mint → Mint

wrap

��

Intuitively, the worker will apply a success continuation of type Int → Mint if
the result of evaluating the given expression is an integer, or a failure continuation
of type Mint (which can be viewed as a function with no arguments) if the result
is an exception. The necessary conversion functions can be defined as follows:

unwrap g e s f = case (g e) of
Nothing → f
Just n → s n

wrap h e = h e Just Nothing

Verifying that wrap ◦ unwrap = idExpr→Mint is straightforward:

(wrap ◦ unwrap) g e
= { applying ◦ }

wrap (unwrap g) e
= { applying wrap }

unwrap g e Just Nothing
= { applying unwrap }

case (g e) of
Just n → Just n
Nothing → Nothing

= { case elimination }
g e

Now that we have established the preconditions for the worker/wrapper trans-
formation, applying this transformation gives:

eval :: Expr → Mint
eval = wrap work

work :: Expr → (Int → Mint) → Mint → Mint
work = unwrap (body (wrap work))

Making the expression argument explicit in the definition for eval , and expanding
out wrap, we obtain the wrapper for the continuation-passing version of eval , which
simply supplies Just and Nothing as the initial continuations:

eval :: Expr → Mint
eval e = work e Just Nothing

Now let us focus our attention on simplifying the definition for the worker func-

tion. First of all, we redefine work using pattern matching

work (Val n) s f = s n
work (Add x y) s f = case wrap work x of

Nothing → f
Just n → case wrap work y of

Nothing → f
Just m → s (n + m)

work (Throw ) s f = f
work (Catch x y) s f = case wrap work x of

Nothing → case wrap work y of
Nothing → f
Just n → s n

Just n → s n

which transformation can be verified as follows:

work e s f
= { applying work }

unwrap (body (wrap work )) e s f
= { applying unwrap }

case (body (wrap work ) e) of
Nothing → f
Just n → s n

= { applying body }
case (case e of

Val n → Just n
Add x y → case wrap work x of

Nothing → Nothing
Just n → case wrap work y of

Nothing → Nothing
Just m → Just (n + m)

Throw → Nothing
Catch x y → case wrap work x of

Nothing → wrap work y
Just n → Just n ) of

Nothing → f
Just n → s n

= { case of case }
case e of

Val n → s n
Add x y → case wrap work x of

Nothing → f
Just n → case wrap work y of

Nothing → f
Just m → s (n + m)

Throw → f
Catch x y → case wrap work x of

Nothing → case wrap work y of
Nothing → f
Just n → s n

Just n → s n

Now we simplify the case analyses used in the recursive definition by exposing
the implicit use of unwrap and then fusing again with wrap. In particular, for an
arbitrary expression x , and continuations g and s of the appropriate types:

case wrap work x of
Nothing → g
Just n → s n

= { unapplying unwrap }
unwrap (wrap work ) x s g

= { worker/wrapper fusion }
work x s g

Applying this result three times gives our final definition for the worker, in which
control flow is now managed by explicit success and failure continuations, rather
than by repeated tagging and untagging of Maybe values:

work :: Expr → (Int → Mint) → Mint → Mint
work (Val n) s f = s n
work (Add x y) s f = work x (λn → work y (λm → s (n + m)) f ) f
work (Throw ) s f = f
work (Catch x y) s f = work x s (work y s f )

Once again, note that everything follows in a straightforward manner once we
have made the initial design decision to represent the type Mint using two con-
tinuations, and that the derivation does not require the use of induction. We con-
clude this section by noting that the resulting worker function can now readily be
transformed into an abstract machine for evaluating expressions (Ager et al., 2003;
Hutton & Wright, 2006), resulting in an efficient machine that operates using two
control stacks, one for normal evaluation and the other for handling exceptions.

8 Related work

As discussed in the introductory section, the worker/wrapper transformation was
first used to exploit strictness information in the Glasgow Haskell Compiler (Pey-
ton Jones & Launchbury, 1991). However, the correctness of the resulting transfor-
mation has never been formally verified. During our work on this article we learned
that the authors did in fact sketch a proof using the rolling rule, but this proof
was never fully developed, nor was it published. The underlying strictness analyser
in this compiler has been changed and improved several times since its initial im-

plementation (Peyton Jones & Partain, 1993), but the use of the worker/wrapper
technique to exploit this information remains exactly the same.

The worker/wrapper transformation has also been used to improve the shortcut
approach to deforestation (Gill et al., 1993), an optimisation technique that at-
tempts to remove intermediate data structures from programs. In particular, the
shortcut approach performs fusion when functions that produce and consume lists
are written in a specific, idiomatic way, and are completely inlined. In the final
chapter of his PhD thesis (1996), Gill describes a scheme for abstracting a function
that produces a list in a way that utilised the worker/wrapper transformation to re-
duce the increase in code size that can result from wholesale inlining of definitions.
In particular, Gill considered list producing functions of the form

f x1 · · · xn = build (λc n → body)

which were then split into a wrapper and a worker

f x1 · · · xn = build (λc n → work x1 · · · xn c n)

work x1 · · · xn c n = body

and finally, the new definition for f was inlined in the definition for work , thereby
communicating the deforestation opportunity. However, no formal justification was
given for the correctness of the worker/wrapper technique.

Chitil considerably extended Gill’s work in (2000b; 2000a), by generalising to
allow for multiple return lists, as directed by his type-inference based approach to
deforestation. In particular, Chitil uses a static argument transformation (Santos,
1995) to turn a recursive function into a non-recursive function with local recur-
sion, and splits the new, non-recursive function into a worker and wrapper, with
detailed justification. He also explores the recursive instance of the worker/wrapper
technique as applied to deforestation, acknowledging that “whereas intuitively the
semantic correctness of [this approach] is clear, a formal proof is hard”, and suggests
that a proof may be possible using improvement theory (Sands, 1998). However, as
we have shown in this article, simpler techniques suffice.

Another application of the worker/wrapper transformation has been in changing
the order of function arguments, to aid static analysis. In particular, Launchbury
and Sheard (1995) use the technique to illustrate how the standard definition

map :: (a → b) → [a ] → [b ]
map f xs = case xs of

[ ] → [ ]
(x : xs) → f x : map f xs

can be translated into an equivalent version that is defined using a worker that

takes the two arguments in the opposite order:

map :: (a → b) → [a ] → [b ]
map f xs = work xs f

work :: [a ] → (a → b) → [b ]
work xs f = case xs of

[ ] → [ ]
(x : xs) → f x : work xs f

It is interesting to note that the various users of the worker/wrapper transforma-
tion appear to realise its potential as a powerful tool for communicating a change
of type through a recursive definition, but did not explore this generalisation.

The worker/wrapper terminology has also been used to describe a similar but dis-
tinct technique, namely when a wrapper of one type is exposed to an optimisation
system for the purpose of applying a type conversion to a worker. Technically, this
corresponds to the adapter design pattern (Gamma et al., 1995). The rewrite sys-
tem of the Glasgow Haskell Compiler (Peyton Jones et al., 2001) uses this technique
in its implementation of short-cut deforestation, as does the recent implementation
of stream fusion (Coutts et al., 2007). In both these systems, rather than wholesale
inlining of a candidate function, an inlined wrapper expresses the opportunity for
deforestation, but the actual computation is performed by the worker. In a real
sense, these uses of workers and wrappers are part of the worker/wrapper transfor-
mation, which is the combination of splitting a function into a worker and wrapper
in a correct and potentially useful manner, together with the use of these new
functions to fulfill specific roles inside a rewrite system.

9 Conclusion and further work

In this article we presented the worker/wrapper transformation, and gave several
examples of its use as an equational reasoning technique for improving the per-
formance of functional programs. In particular, we showed how it can be used to
derive more efficient programs based upon the use of accumulation, unboxed val-
ues, memoisation, and continuations. Since these are all well studied and widely
used approaches to program optimisation, it is natural to ask what new value the
worker/wrapper transformation brings? We see three primary benefits:

• It provides a general and systematic approach to transforming a computa-
tion of one type into an equivalent computation of another type. Such type
transformations are pervasive in functional programming and reasoning, and
in the efficient compilation of functional languages.

• It is straightforward to understand and apply, requiring only basic equational
reasoning techniques, and often avoiding the need for induction. As such, it
can readily be utilised by a wide spectrum of functional programmers as a
simple but powerful technique for refactoring their programs.

• It captures many seemingly unrelated optimisation techniques inside a single

unified framework, which may help reveal new connections between existing
techniques, and the discovery of new techniques.

This article is by no means the end of the worker/wrapper story, but rather
opens up a number of opportunities for further research. Interesting topics for fur-
ther work include mechanising the technique in an equational reasoning system
such as HERA (Gill, 2006), investigating how it can be automated in an opti-
mising compiler such as GHC (Peyton Jones et al., 2001), considering programs
that utilise effects (Wadler, 1992a; McBride & Paterson, 2008) and more advanced
applications (e.g. to programming languages themselves), exploring the use of spe-
cialised patterns of recursion (Hutton, 1999; Gibbons & Jones, 1998) and versions
of the transformation (e.g. for changing the result type of a function), and further
generalising the technique using category theory (Backhouse et al., 1995).

Appendix

For completeness, this appendix briefly reviews the necessary semantic theory of
fixed points that underlies our formalisation of the worker/wrapper transformation,
and presents a proof of the rolling rule in this context.

Our formalisation is based upon the domain-theoretic approach to semantics
(Schmidt, 1986), in which the basic idea is that types are complete partial orders
(cpos), that is, sets with a partial-ordering � , a least element ⊥, and limits of all
non-empty chains. In turn, programs are continuous functions , that is, functions
between cpos that preserve the partial-order and limit structure.

Now consider a recursive equation x = f x that defines a value x in terms of
itself and some continuous function f . A well-known fixed point theorem states
that this equation has a least solution for x , denoted by fix f and called the least
fixed point of f , which is adopted as the semantics of the definition. Moreover, fix f
is constructed as the limit of the following infinite chain:

⊥ � f ⊥ � f (f ⊥) � f (f (f ⊥)) � · · ·
As a simple example of this approach, consider the equation ones = 1 :ones that

defines the infinite list 1:1:1: · · ·. This definition can be rewritten as ones = f ones ,
where f is the function defined by f xs = 1:xs. Hence, the semantics of the definition
is given by ones = fix f , and by the fixed point theorem is constructed as the limit
of the infinite chain of partial lists containing increasing numbers of 1s:

⊥ � 1 : ⊥ � 1 : 1 : ⊥ � 1 : 1 : 1 : ⊥ � · · ·
The fixed point approach to recursive definitions can be realised in Haskell by

defining an explicit version of the function fix as follows:

fix :: (a → a) → a
fix f = f (fix f )

For example, evaluating the expression fix (λxs → 1 : xs) using this definition gives
the infinite list of ones, 1 : 1 : 1 : · · ·, as expected.

In order to prove the rolling rule, which states that fix (g ◦ f ) = g (fix (f ◦ g))
for any functions f and g of the appropriate types, we exploit the following two
fundamental properties of fixed points (Backhouse, 2002):

fix f = f (fix f ) (computation)

f x � x ⇒ fix f � x (induction)

The first property states that fix f is a fixed point of f (an expression x satisfying
f x = x ), while the second states that fix f is the least prefix point of f (the least
expression x satisfying f x � x ). One might have expected in the latter case to
state that fix f is the least fixed point rather than just the least prefixed point, but
these two notions can be shown to be equivalent, and the above formulation has the
advantage of being more useful for reasoning purposes. Using these two properties,
the rolling rule can now be verified by mutual inclusion:

fix (g ◦ f ) � g (fix (f ◦ g))
⇐ { induction }

(g ◦ f ) (g (fix (f ◦ g))) � g (fix (f ◦ g))
⇔ { applying ◦ }

g (f (g (fix (f ◦ g)))) � g (fix (f ◦ g))
⇔ { unapplying ◦ }

g ((f ◦ g) (fix (f ◦ g))) � g (fix (f ◦ g))
⇔ { computation }

g (fix (f ◦ g)) � g (fix (f ◦ g))
⇔ { reflexivity }

True

and

g (fix (f ◦ g)) � fix (g ◦ f )
⇔ { computation }

g (fix (f ◦ g)) � (g ◦ f ) (fix (g ◦ f ))
⇔ { applying ◦ }

g (fix (f ◦ g)) � g (f (fix (g ◦ f )))
⇐ { g is monotonic }

fix (f ◦ g) � f (fix (g ◦ f ))
⇔ { above result, swapping f and g }

True

We conclude this appendix by noting that the above proof of the rolling rule
only relies on the basic notion of monotonic functions on partially-ordered sets,
rather than the stronger notion of continuous functions on cpo’s. However, working
in this stronger setting automatically guarantees that the necessary fixed points
always exist, by virtue of the fixed point theorem described above, which in weaker
settings may become a side condition on the rolling rule.

